

# ComputerMathematik - Einführung in Maple / Teil IV

Winfried Auzinger (Mai 2009)

Teil IV stellt eines der wichtigsten Packages vor, nämlich

**LinearAlgebra** für [ Numerische ] Lineare Algebra,

plus Beispiele und weitere Ergänzungen.

## 1. Vektoren und Matrizen

Für den Bereich der Linearen Algebra verwendet Maple zwei Varianten des Array-Typs (intern basierend auf **rtables** = 'rectangular tables'): **Vector** und **Matrix**.

ACHTUNG... keine 'last-name-evaluation' (wie z.B. bei tables) - siehe Beispiel:

```
> restart;
> ? rtable
> t1:=table([[1,b]]); t2:=t1; t2; eval(t2);
      t1 := table([1 = [1, b]])
      t2 := t1
      t1
      table([1 = [1, b]])
> rt1:=rtable([[a,b],[1]]); rt2:=t1;
      rt1 :=  $\begin{bmatrix} a & b \\ 1 & 0 \end{bmatrix}$ 
      rt2 := t1
```

**rtables** sind die 'Basis-Datenstruktur' für Arrays, Vektoren und Matrizen, etc. (mit variabler interner Speicherorganisation, z.B. für symmetrische, datendünne, ... etc. Fälle). In der Praxis verwendet man **Array**, für Lineare Algebra insbesondere **Vector** und **Matrix** (Indizierung beginnt fix mit 1).

(Anmerkung: Listen und tables sind für algebraische Berechnungen im allgemeinen nicht gut geeignet.)

```
> x:=Vector([a,b,c]); # Vector() erwartet Liste
      x :=  $\begin{bmatrix} a \\ b \\ c \end{bmatrix}$ 
> whattype(x), type(x,rtable);
      Vector_column, true
```

Es werden also (ähnlich wie in MATLAB) Zeilen- und Spaltenvektoren unterschieden, gemäß dem üblichen Matrix-Vektor-Kalkül.

Default = Spaltenvektor (column)

```
> x:=Vector[column]([a,b,c]); y:=Vector[row]([1,2,3]);
```

$$x := \begin{bmatrix} a \\ b \\ c \end{bmatrix}$$

$$y := [ 1 \ 2 \ 3 ]$$

Alternative Syntax:

> **x:=<1,2,3>; # , steht für 'neue Zeile'**

$$x := \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$$

> **y:=<alpha|beta|delta>; # | steht für 'neue Spalte'**

$$y := [ \alpha \ \beta \ \delta ]$$

Matrix-Typ mit analoger Syntax (Spezifikation mittels listlist):

> **A:=Matrix([[a,b,c],  
[d,e,f]]); # listlist wird zeilenweise interpretiert**

$$A := \begin{bmatrix} a & b & c \\ d & e & f \end{bmatrix}$$

Alternative Syntax:

> **A:=<<a|b|c>,<d|e|f>>; # zeilenweise**

$$A := \begin{bmatrix} a & b & c \\ d & e & f \end{bmatrix}$$

> **A:=<<a,d>|<b,e>|<c,f>>; # spaltenweise**

$$A := \begin{bmatrix} a & b & c \\ d & e & f \end{bmatrix}$$

Vektorelement, Teilvektor:

> **x, x[2], x[1..2];**

$$\begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}, 2, \begin{bmatrix} 1 \\ 2 \end{bmatrix}$$

Matrizelement, Teilmatrix:

> **A, A[2,3], A[1..2,2..3];**

$$\begin{bmatrix} a & b & c \\ d & e & f \end{bmatrix}, f, \begin{bmatrix} b & c \\ e & f \end{bmatrix}$$

Zeile, Spalte einer Matrix:

> **LinearAlgebra[Row](A,1), LinearAlgebra[Column](A,2);**

$$\begin{bmatrix} a & b & c \end{bmatrix}, \begin{bmatrix} b \\ e \end{bmatrix}$$

-- 'Leere' Objekte erzeugen (Initialisierung mit Nullelementen)

> **Vector(3), Matrix(1..3,2);**

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 \\ 0 & 0 \\ 0 & 0 \end{bmatrix}$$

-- Angabe eines Namens für die Elemente:

> **B:=Matrix(2,5,symbol='b'); B[1,1]=b[1,1]; evalb(%);**

$$B := \begin{bmatrix} b_{1,1} & b_{1,2} & b_{1,3} & b_{1,4} & b_{1,5} \\ b_{2,1} & b_{2,2} & b_{2,3} & b_{2,4} & b_{2,5} \end{bmatrix}$$

$$b_{1,1} = b_{1,1}$$

*true*

> **b[1,1]:=1; B; # b[i,j] nun synonym zu B[i,j];**

$$b_{1,1} := 1$$

$$\begin{bmatrix} 1 & b_{1,2} & b_{1,3} & b_{1,4} & b_{1,5} \\ b_{2,1} & b_{2,2} & b_{2,3} & b_{2,4} & b_{2,5} \end{bmatrix}$$

> **b[1,1], B[1,1];**

1, 1

-- Angabe einer Funktion (i,j)->f(i,j), die die Einträge definiert:

> **f:=(i,j)->i+j; B:=Matrix(2,4,f);**

$$f := (i, j) \rightarrow i + j$$

$$B := \begin{bmatrix} 2 & 3 & 4 & 5 \\ 3 & 4 & 5 & 6 \end{bmatrix}$$

Auch 'Lego spielen' kann man:

> **x; B:=Matrix([[x,x]]);**

$$\begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$$

$$B := \begin{bmatrix} 1 & 1 \\ 2 & 2 \\ 3 & 3 \end{bmatrix}$$

... + weitere syntaktische Varianten...

-- Elementare Arithmetik: (. für MV bzw. MM-Multiplikation, nicht \* !):

> **B,x,B.x[1..2];**

$$\begin{bmatrix} 1 & 1 \\ 2 & 2 \\ 3 & 3 \end{bmatrix}, \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}, \begin{bmatrix} 3 \\ 6 \\ 9 \end{bmatrix}$$

```
> B+2*B; B[1..2,1..2].B[1..2,1..2];
```

$$\begin{bmatrix} 3 & 3 \\ 6 & 6 \\ 9 & 9 \end{bmatrix}$$
$$\begin{bmatrix} 3 & 3 \\ 6 & 6 \end{bmatrix}$$

-- Zu beachten:

Gemäß dem zugrundeliegenden, sehr 'flexiblen' Datentyp **rtable** können Objekten vom Typ **Vector** und **Matrix** verschiedene Eigenschaften zugewiesen werden (z.B. Symmetrie, `sparsity`), die auch auf die Art der Speicherung einen Einfluss haben (später). Siehe z.B.

```
> ? shape
```

```
> ? storage
```

Für nicht zu große Objekte braucht man sich darum i.Allg. nicht weiter zu kümmern.

## 2. Das Paket LinearAlgebra

Bemerkung: Es gibt auch ein älteres Paket **linalg** (weniger umfangreich).

```
> restart;
```

```
> with(LinearAlgebra);
```

```
[&x, Add, Adjoint, BackwardSubstitute, BandMatrix, Basis, BezoutMatrix, BidiagonalForm, BilinearForm, CharacteristicMatrix, CharacteristicPolynomial, Column, ColumnDimension, ColumnOperation, ColumnSpace, CompanionMatrix, ConditionNumber, ConstantMatrix, ConstantVector, Copy, CreatePermutation, CrossProduct, DeleteColumn, DeleteRow, Determinant, Diagonal, DiagonalMatrix, Dimension, Dimensions, DotProduct, EigenConditionNumbers, Eigenvalues, Eigenvectors, Equal, ForwardSubstitute, FrobeniusForm, GaussianElimination, GenerateEquations, GenerateMatrix, Generic, GetResultDataType, GetResultShape, GivensRotationMatrix, GramSchmidt, HankelMatrix, HermiteForm, HermitianTranspose, HessenbergForm, HilbertMatrix, HouseholderMatrix, IdentityMatrix, IntersectionBasis, IsDefinite, IsOrthogonal, IsSimilar, IsUnitary, JordanBlockMatrix, JordanForm, KroneckerProduct, LA_Main, LUdecomposition, LeastSquares, LinearSolve, Map, Map2, MatrixAdd, MatrixExponential, MatrixFunction, MatrixInverse, MatrixMatrixMultiply, MatrixNorm, MatrixPower, MatrixScalarMultiply, MatrixVectorMultiply, MinimalPolynomial, Minor, Modular, Multiply, NoUserValue, Norm, Normalize, NullSpace, OuterProductMatrix, Permanent, Pivot, PopovForm, QRdecomposition, RandomMatrix, RandomVector, Rank, RationalCanonicalForm, ReducedRowEchelonForm, Row, RowDimension, RowOperation, RowSpace, ScalarMatrix, ScalarMultiply, ScalarVector, SchurForm, SingularValues, SmithForm, StronglyConnectedBlocks, SubMatrix, SubVector, SumBasis, SylvesterMatrix, ToeplitzMatrix, Trace, Transpose, TridiagonalForm, UnitVector, VandermondeMatrix, VectorAdd, VectorAngle, VectorMatrixMultiply, VectorNorm, VectorScalarMultiply, ZeroMatrix, ZeroVector, Zip]
```

## 2.1 Grundsätzliches zu Verwendung von LinearAlgebra

```
> A:=Matrix([[1,2,3],[1,a,b]]); Dimension(A);
```

$$A := \begin{bmatrix} 1 & 2 & 3 \\ 1 & a & b \end{bmatrix}$$

2, 3

```
> RowDimension(A), ColumnDimension(A);
```

2, 3

```
> Rank(A); # 'generischer' Rang!
```

2

-- **LinearAlgebra** eignet sich sowohl für symbolische als auch für numerische bzw. gemischte Berechnungen

```
> B:=A.Transpose(A);
```

$$B := \begin{bmatrix} 14 & 1 + 2a + 3b \\ 1 + 2a + 3b & 1 + a^2 + b^2 \end{bmatrix}$$

```
> Determinant(B); B^(-1); # generisch!
```

$$13 + 10a^2 + 5b^2 - 4a - 6b - 12ab$$

$$\begin{bmatrix} \frac{1 + a^2 + b^2}{13 + 10a^2 + 5b^2 - 4a - 6b - 12ab} & -\frac{1 + 2a + 3b}{13 + 10a^2 + 5b^2 - 4a - 6b - 12ab} \\ -\frac{1 + 2a + 3b}{13 + 10a^2 + 5b^2 - 4a - 6b - 12ab} & \frac{14}{13 + 10a^2 + 5b^2 - 4a - 6b - 12ab} \end{bmatrix}$$

```
> C:=Transpose(A).A;
```

$$C := \begin{bmatrix} 2 & 2 + a & 3 + b \\ 2 + a & 4 + a^2 & 6 + ab \\ 3 + b & 6 + ab & 9 + b^2 \end{bmatrix}$$

```
> Determinant(C); C^(-1); # diese Matrix ist 'generisch singular'  
# (unabh. von a,b)
```

0

Error, (in rtable/Power) singular matrix

-- Für **numerische Berechnungen** gibt es zwei Varianten

```
> UseHardwareFloats;
```

*deduced*

Der Wert der Umgebungsvariablen **UseHardwareFloats** steuert gemeinsam mit der eingestellten Rechengenauigkeit **Digits** das Verhalten:

```
> UseHardwareFloats:=false; Digits:=20;
```

*UseHardwareFloats := false*

*Digits := 20*

```
> x:=Vector([1/3.0,1/2.0]);
```

$$x := \begin{bmatrix} 0.33333333333333333333 \\ 0.50000000000000000000 \end{bmatrix}$$

```
> x.x; # inneres Produkt auf 20 Stellen genau berechnet
0.36111111111111111111
```

(Aber Achtung: # der angezeigten Stellen (hier: 20) einzustellen über **Tools/Options**

```
> UseHardwareFloats:=true;
UseHardwareFloats := true
```

Dies bedeutet, dass die maschineninterne **doppelt genaue IEEE-Arithmetik** wird (binär, ca. 16 Dezimalstellen, wie in MATLAB!)

```
> x:=Vector([1/3.0,1/2.0]);
x := [ 0.33333333333333333333
       0.50000000000000000000 ]
```

Oder konsequenter auch mit 'IEEE-Daten':

(**evalhf** steht für 'evaluate hardware float' (= IEEE double), nicht zu verwechseln mit evalf)

(Beachte den Unterschied zu vorher - weniger als die 20 angezeigten Stellen sind signifikant:)

```
> x:=Vector([evalhf(1/3),evalhf(sqrt(2))]);
x := [ 0.33333333333333331500
       1.41421356237309515000 ]
```

Besser so schreiben (geht nur mit numerischen Daten!):

```
> x:=map(evalhf,Vector([1/3,sqrt(2)]));
x := [ 0.33333333333333331500
       1.41421356237309515000 ]
```

-- Zur Kontrolle, was intern passiert:

```
> infolevel[LinearAlgebra]:=1;
infolevelLinearAlgebra := 1
```

```
> x.x; # hier werden 'Hardware-Routinen' aktiviert
DotProduct: calling external function
DotProduct: NAG hw_f06eaf
2.11111111111111116000
```

```
> UseHardwareFloats:=false;
UseHardwareFloats := false
```

```
> y:=Vector([1.0,2.0]); y.y;
# hier werden 'Software-Routinen' aktiviert
y := [ 1.00000000000000000000
       2.00000000000000000000 ]
```

```
DotProduct: calling external function
DotProduct: NAG sw_f06eaf
5.00000000000000000000
```

ANMERKUNG: In Prozeduren kann man generell mittels **option hfloat** die Verwendung der Hardware-Arithmetik erzwingen.

```
> restart: with(LinearAlgebra): UseHardwareFloats;
# default deduced='automatisch'
```

## 2.2 Symbolisches und exaktes Lösen von Problemen

(anhand von Beispielen)

-- Ein lineares Gleichungssystem:

```
> n:=5; A:=Matrix(n,n,(i,j)->alpha*i-j);
      n := 5
```

$$A := \begin{bmatrix} \alpha - 1 & \alpha - 2 & \alpha - 3 & \alpha - 4 & \alpha - 5 \\ 2\alpha - 1 & 2\alpha - 2 & 2\alpha - 3 & 2\alpha - 4 & 2\alpha - 5 \\ 3\alpha - 1 & 3\alpha - 2 & 3\alpha - 3 & 3\alpha - 4 & 3\alpha - 5 \\ 4\alpha - 1 & 4\alpha - 2 & 4\alpha - 3 & 4\alpha - 4 & 4\alpha - 5 \\ 5\alpha - 1 & 5\alpha - 2 & 5\alpha - 3 & 5\alpha - 4 & 5\alpha - 5 \end{bmatrix}$$

```
> b:=Vector([seq(i,i=1..n)]);
```

$$b := \begin{bmatrix} 1 \\ 2 \\ 3 \\ 4 \\ 5 \end{bmatrix}$$

```
> LinearSolve(A,b); # i.Allg. 3-dimensionale Lösungsschar
# mit freien Parametern
```

$$\begin{bmatrix} \frac{-t_3 \alpha + 2 - t_4 \alpha + 3 - t_5 \alpha + 2}{\alpha} \\ -\frac{1 + 2 - t_3 \alpha + 3 - t_4 \alpha + 4 - t_5 \alpha}{\alpha} \\ -t_3 \\ -t_4 \\ -t_5 \end{bmatrix}$$

```
> simplify(subs(_t[1]=1,%));
```

$$\begin{bmatrix} \frac{-t_3 \alpha + 2 - t_4 \alpha + 3 - t_5 \alpha + 2}{\alpha} \\ -\frac{1 + 2 - t_3 \alpha + 3 - t_4 \alpha + 4 - t_5 \alpha}{\alpha} \\ -t_3 \\ -t_4 \\ -t_5 \end{bmatrix}$$

```
> alpha:=0; A;
```

$\alpha := 0$

$$\begin{bmatrix} -1 & -2 & -3 & -4 & -5 \\ -1 & -2 & -3 & -4 & -5 \\ -1 & -2 & -3 & -4 & -5 \\ -1 & -2 & -3 & -4 & -5 \\ -1 & -2 & -3 & -4 & -5 \end{bmatrix}$$

> LinearSolve(A,b); # jetzt ist die Matrix singular

Error, (in LinearAlgebra:-LA\_Main:-LinearSolve) inconsistent system

> n:=20; A:=Matrix(n,n); A[1,1];

n := 20

$$A := \begin{bmatrix} 20 \times 20 \text{ Matrix} \\ \text{Data Type: anything} \\ \text{Storage: rectangular} \\ \text{Order: Fortran\_order} \end{bmatrix}$$

0

> A^(-1);

Error, (in rtable/Power) singular matrix

-- Eigenproblem und Jordan'sche Normalform:

> A:=Matrix([[c,a,c],[d,c,d],[c,d,c]]);

$$A := \begin{bmatrix} c & a & c \\ d & c & d \\ c & d & c \end{bmatrix}$$

> Eigenvalues(A);

$$\begin{bmatrix} 0 \\ \frac{3}{2}c + \frac{1}{2}\sqrt{c^2 + 4da + 4d^2} \\ \frac{3}{2}c - \frac{1}{2}\sqrt{c^2 + 4da + 4d^2} \end{bmatrix}$$

> Eigenvectors(A): %[1];

$$\begin{bmatrix} \frac{3}{2}c + \frac{1}{2}\sqrt{c^2 + 4da + 4d^2} \\ \frac{3}{2}c - \frac{1}{2}\sqrt{c^2 + 4da + 4d^2} \\ 0 \end{bmatrix}$$

> J:=JordanForm(A); # immer diagonalisierbar,

# ggfs. mit konj.komplexen Eigenwerten



$$J := \begin{bmatrix} 0 & 0 & 0 \\ 0 & \frac{3}{2}c - \frac{1}{2}\sqrt{c^2 + 4da + 4d^2} & 0 \\ 0 & 0 & \frac{3}{2}c + \frac{1}{2}\sqrt{c^2 + 4da + 4d^2} \end{bmatrix}$$

**> X:=JordanForm(A,output='Q');**

$$X := \left[ \left[ \begin{array}{c} -\frac{c^2 - d^2}{-2c^2 + d^2 + da}, \\ \frac{1}{2} \frac{-c^2 \sqrt{c^2 + 4da + 4d^2} + da \sqrt{c^2 + 4da + 4d^2} + cda + c^3 - 2d^2 c}{\sqrt{c^2 + 4da + 4d^2} (-2c^2 + d^2 + da)}, \\ \frac{1}{2} \frac{-c^3 - c^2 \sqrt{c^2 + 4da + 4d^2} - cda + da \sqrt{c^2 + 4da + 4d^2} + 2d^2 c}{\sqrt{c^2 + 4da + 4d^2} (-2c^2 + d^2 + da)} \end{array} \right], \right. \\ \left. \left[ 0, -\frac{d}{\sqrt{c^2 + 4da + 4d^2}}, \frac{d}{\sqrt{c^2 + 4da + 4d^2}} \right], \right. \\ \left. \left[ \frac{c^2 - d^2}{-2c^2 + d^2 + da}, -\frac{1}{2} \frac{c^2 \sqrt{c^2 + 4da + 4d^2} - d^2 \sqrt{c^2 + 4da + 4d^2} - c^3 + 2cda - d^2 c}{\sqrt{c^2 + 4da + 4d^2} (-2c^2 + d^2 + da)}, \right. \right. \\ \left. \left. -\frac{1}{2} \frac{c^3 + c^2 \sqrt{c^2 + 4da + 4d^2} + d^2 c - d^2 \sqrt{c^2 + 4da + 4d^2} - 2cda}{\sqrt{c^2 + 4da + 4d^2} (-2c^2 + d^2 + da)} \right] \right]$$

Anderes Beispiel:

**> A:=Matrix([[1,epsilon],[0,1]]);**

$$A := \begin{bmatrix} 1 & \epsilon \\ 0 & 1 \end{bmatrix}$$

**> JordanForm(A); # i.Allg. nicht diagonalisierbar  
# (offenbar für epsilon<>0)**

$$\begin{bmatrix} 1 & 1 \\ 0 & 1 \end{bmatrix}$$

ABER: Für epsilon=0 ist A diagonalisierbar!

**> epsilon:=0; JordanForm(A);**

$$\epsilon := 0$$

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

**> SingularValues(HilbertMatrix(2));**

$$\begin{bmatrix} \frac{2}{3} + \frac{1}{6}\sqrt{13} \\ \frac{2}{3} - \frac{1}{6}\sqrt{13} \end{bmatrix}$$

-- Einige weitere Transformationen der Linearen Algebra:

```
> V:=VandermondeMatrix([xi[1],xi[2],xi[3]]);
```

$$V := \begin{bmatrix} 1 & \xi_1 & \xi_1^2 \\ 1 & \xi_2 & \xi_2^2 \\ 1 & \xi_3 & \xi_3^2 \end{bmatrix}$$

```
> HessenbergForm(V); # das geht nicht symbolisch,  
# der Algorithmus benötigt  
# explizite numerische Vergleiche!
```

Error, (in LinearAlgebra:-LA\_Main:-HessenbergForm) Matrix does not evaluate to floating point

Aber:

```
> V:=VandermondeMatrix([1,2,3]);
```

$$V := \begin{bmatrix} 1 & 1 & 1 \\ 1 & 2 & 4 \\ 1 & 3 & 9 \end{bmatrix}$$

```
> HessenbergForm(V); # rechnet automatisch in floating point  
# rechts oben steht eigentlich 0
```

$$\begin{bmatrix} 1.00000000000000000000 & -1.41421356237309504880 & 2.00000000000000000000 \cdot 10^{-30} \\ -1.41421356237309504880 & 9.00000000000000000000 & -4.00000000000000000000 \\ 0.00000000000000000000 & -3.00000000000000000000 & 2.00000000000000000000 \end{bmatrix}$$

```
> UseHardwareFloats:=true;
```

*UseHardwareFloats := true*

```
> HessenbergForm(V); # double-Version
```

$$\begin{bmatrix} 1.00000000000000000000 & -1.41421356237309470000 & 2.02745806254789330000 \cdot 10^{-17} \\ -1.41421356237309515000 & 8.99999999999999644000 & -3.99999999999999912000 \\ 0.00000000000000000000 & -3.00000000000000000000 & 1.9999999999999978000 \end{bmatrix}$$

```
> H:=HilbertMatrix(3);
```

$$H := \begin{bmatrix} 1 & \frac{1}{2} & \frac{1}{3} \\ \frac{1}{2} & \frac{1}{3} & \frac{1}{4} \\ \frac{1}{3} & \frac{1}{4} & \frac{1}{5} \end{bmatrix}$$

Orthogonalisierung der Spalten (geht auch exakt):

```
> GramSchmidt([seq(Column(H,i),i=1..RowDimension(H))]);
```

$$\begin{bmatrix} \begin{bmatrix} 1 \\ \frac{1}{2} \\ \frac{1}{3} \end{bmatrix}, \begin{bmatrix} -\frac{5}{98} \\ \frac{17}{294} \\ \frac{13}{196} \end{bmatrix}, \begin{bmatrix} \frac{1}{2190} \\ -\frac{1}{365} \\ \frac{1}{365} \end{bmatrix} \end{bmatrix}$$

Matrix-Exponentialfunktion (basiert auf Jordan-Zerlegung);

```
> C:=CompanionMatrix('x^2+rho*x','x');
```

$$C := \begin{bmatrix} 0 & 0 \\ 1 & -\rho \end{bmatrix}$$

```
> MatrixExponential(C); # offenbar ist rho=0 Sonderfall!
```

$$\begin{bmatrix} 1 & 0 \\ -\frac{e^{-\rho}-1}{\rho} & e^{-\rho} \end{bmatrix}$$

QR-Zerlegung (verwandt mit Gram-Schmidt):

```
> QRDecomposition(C);
```

$$\begin{bmatrix} 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 1 & -\rho \end{bmatrix}$$

... und ... und ...

Grundprinzip: Symbolisch exakte Lösungen werden versucht, wenn der zugrunde liegende Algorithmus dies grundsätzlich zulässt.

Numerische Ergebnisse in exakter Arithmetik möglich, aber manche Algorithmen, die nur in der Numerischen linearen Algebra verwendet werden, sind nur als Gleitpunkt-Versionen implementiert (z.B. HessenbergForm).

**Kompetente Verwendung erfordert teilweise Kenntnisse über [Numerische] Lineare Algebra.**  
**UE: Ggf. werden Hinweise zu den Beispielen gegeben.**

```
> restart: with(LinearAlgebra): UseHardwareFloats:=true:
```

## ▼ 2.3 Numerische Lineare Algebra

Mit der Einstellung **UseHardwareFloats:=true** bietet Maple/LinearAlgebra einen Funktionsumfang wie MATLAB, teilweise auch darüber hinaus; Implementierung teilweise weniger effizient (?) - für 'nicht zu große' Probleme egal.

EinigeBeispiele:

```
-- Eigenproblem für eine symmetrische 20x20 - Matrix:
```

```
> H:=map(evalhf,HilbertMatrix(20)); H[1..3,1..3];
```

$$H := \begin{bmatrix} 20 \times 20 \text{ Matrix} \\ \text{Data Type: anything} \\ \text{Storage: rectangular} \\ \text{Order: Fortran\_order} \end{bmatrix}$$

```

[ 1.00000000000000000000000000000000 0.50000000000000000000000000000000 0.33333333333333331500
  0.50000000000000000000000000000000 0.33333333333333331500 0.25000000000000000000000000000000
  0.33333333333333331500 0.25000000000000000000000000000000 0.20000000000000000000000000000000 ]

```

```
> Eigenvalues(H);%[1];
```

```

[ 1 .. 20 Vectorcolumn
  Data Type: complex8
  Storage: rectangular
  Order: Fortran_order ]

```

```
1.90713472040725351000 + 0.00000000000000000000000000000000 I
```

```
> JordanForm(H); # nicht sinnvoll definiert!
```

```
Error, (in LinearAlgebra:-LA_Main:-JordanForm) Jordan form not defined for a Matrix with floating-point values in its entries
```

```
> ConditionNumber(evalhf(0.1)*IdentityMatrix(20)+H);
```

```
68.97878013078577769165
```

```
-- Berechnung einer Ausgleichsgeraden, mit Grafik:
```

```
> n:=7;
```

```
n := 7
```

```
> t:=seq(i,i=1..n); x:=map(evalhf,[seq(i^3,i=1..n)]);
```

```
t := [1, 2, 3, 4, 5, 6, 7]
```

```
x := [1.00000000000000000000000000000000, 8.00000000000000000000000000000000, 27.00000000000000000000000000000000,
64.00000000000000000000000000000000, 125.00000000000000000000000000000000, 216.00000000000000000000000000000000,
343.00000000000000000000000000000000]
```

```
> V:=VandermondeMatrix(t,n,2);
```

```

V :=
[ 1 1
  1 2
  1 3
  1 4
  1 5
  1 6
  1 7 ]

```

```
> X:=convert(x,Vector);
```

```


X :=
[ 1.00000000000000000000000000000000
  8.00000000000000000000000000000000
  27.00000000000000000000000000000000
  64.00000000000000000000000000000000
  125.00000000000000000000000000000000
  216.00000000000000000000000000000000
  343.00000000000000000000000000000000 ]

```

```

> LinearSolve(V,X);
Error, (in LinearAlgebra:-LA_Main:-LinearSolve) inconsistent system
> y:=LeastSquares(V,X); # löst Ausgleichsproblem
      y :=  $\begin{bmatrix} -108.000000000000000000000000 \\ 55.000000000000000000000000 \end{bmatrix}$ 
> gerade:=tau->y[1]+tau*y[2];
      gerade :=  $\tau \rightarrow y_1 + \tau y_2$ 
> p1:=plot(gerade(tau),tau=t[1]-1..t[n]+1,thickness=6,color=blue):
> p2:=plots[pointplot]([seq([t[i],x[i]],i=1..n)],
      symbolsize=20,symbol=circle,color=red):
> plots[display](p1,p2,title="Ausgleichsgerade");

```



### ▼ 3. Programmierbeispiele

#### ▼ 3.1 Ein allgemeineres lineares Ausgleichsproblem

Gegeben sei eine beliebige Funktion

```
> x:='x': f:=x->'f'(x); type(f,procedure), f(x);  
f:=x→f'(x)  
true,f(x)
```

und ein approximierender Ansatz für f als Polynom vom Grad n:

```
> p:=(x,n)->add(c[i]*x^i,i=0..n);  
p:=(x,n)→add(cixi,i=0..n)
```

Die Parameter c[i] sind so zu bestimmen, dass folgendes Integral minimal wird:

```
> Int((p(xi,n)-f(xi))^2,xi=a..b);  

$$\int_a^b (c_0 + c_1 \xi + c_2 \xi^2 + c_3 \xi^3 + c_4 \xi^4 + c_5 \xi^5 + c_6 \xi^6 + c_7 \xi^7 - f(\xi))^2 d\xi$$

```

Wir schreiben dafür eine Prozedur, die die Minimalbedingung

'partielle Ableitungen nach den c[i] müssen 0 sein'

mittels LinearAlgebra implementiert:

```
> minimiere:=proc(f::procedure,a,b,n)  
# hier fehlen eigentlich noch die description  
# und ein Kommentar zur Bedeutung der lokalen Variablen...  
local A,ableitung,c,i,k,p,r,sol,zielfunktional;  
p:=xi->add(c[i]*xi^i,i=0..n);  
zielfunktional:=int((p(xi)-f(xi))^2,xi=a..b);  
# partiell nach den c[i] differenzieren:  
for i from 0 to n do  
ableitung[i]:=simplify(diff(zielfunktional,c[i]));  
# printf("\n ableitung[%d] = %a",i,ableitung[i]);  
end do;  
# Koeffizientenvergleich in allen Ableitungen  
# nach allen c[k], in Matrix A speichern:  
# Achtung: Indizierung muss in A mit 1 beginnen  
A:=Matrix(n+1,n+1);  
for i from 0 to n do  
for k from 0 to n do  
A[i+1,k+1]:=coeff(ableitung[i],c[k]);  
end do;  
end do;  
# In den Ableitungen sind auch Terme enthalten,  
# in denen die c[k] nicht vorkommen:  
r:=Vector(n+1);  
for i from 0 to n do  
r[i+1]:=subs(seq(c[k]=0,k=0..n),ableitung[i]);  
end do;  
# Lösen des Gleichungssystems A.c+r=0,  
# zurückspeichern in die c[i] (Indizierung mit 0 beginnend)  
sol:=LinearSolve(A,-r);  
for i from 0 to n do  
c[i]:=sol[i+1];  
end do;
```

```

# Zurückgegeben wird das bestapproximierende
# Polynom als Ausdruck in der Variablen xi
# und der Wert des Zielfunktional (nochmals ausgewertet)
zielfunktional:=int((p(xi)-f(xi))^2,xi=a..b);
return collect(expand(p(xi)),xi),zielfunktional;
end proc:

```

Ein erster Test: Beliebiges Polynom vom Grad n reproduziert sich selbst:

```
> n:=4; minimiere(x->sum('u[i]*x^i','i'=0..n),a,b,n);
```

$n := 4$

$$u_0 + u_1 \xi + u_2 \xi^2 + u_3 \xi^3 + u_4 \xi^4, 0$$

Approximation eines Polynoms vom Grad 2 auf [0,1] durch eine Gerade:

```
> minimiere(x->u[0]+u[1]*x+u[2]*x^2,0,1,1);
```

$$-\frac{1}{6} u_2 + u_0 + (u_2 + u_1) \xi, \frac{1}{180} u_2^2$$


Eine numerische Rechnung: Approximation für sin

```
> p,minimum:=minimiere(sin,0.0,Pi,2);
```

$p, minimum := -0.05046549777844937940 + 1.31223620483245385000 \xi$

$$- 0.41769775700646794000 \xi^2, 0.00093629432592514968$$

```
> plot([p,sin(xi)],xi=0..Pi,thickness=4);
```



### 3.2 Eine einfache Matrix-Rekursion:

Berechnung von  $A^{(2^n)} = A^{(2^{(n-1)})}.A^{(2^{(n-1)})} = \dots$

```
> restart: with(LinearAlgebra):
> Apower:=proc(A::Matrix,n::nonnegint)
> description "Rekursive Berechnung von A^(2^n)";
> local B;
> if (RowDimension(A)<>ColumnDimension(A)) then
>   error "A muss quadratische Matrix sein"
> end if;
> if n=0 then
>   return A
> else
>   B:=Apower(A,n-1);
>   return B.B
> end if;
> end proc;
> M:=Matrix([[1,2],[3,4]]);
```



$$M := \begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix}$$

> Apower(M,0);

$$\begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix}$$

> Apower(M,1), M^2;

$$\begin{bmatrix} 7 & 10 \\ 15 & 22 \end{bmatrix}, \begin{bmatrix} 7 & 10 \\ 15 & 22 \end{bmatrix}$$

> Apower(M,5); M^(2^5);

$$\begin{bmatrix} 55368923250321279115351 & 80696169672298405404130 \\ 121044254508447608106195 & 176413177758768887221546 \end{bmatrix}$$

$$\begin{bmatrix} 55368923250321279115351 & 80696169672298405404130 \\ 121044254508447608106195 & 176413177758768887221546 \end{bmatrix}$$

> Apower(Matrix([[a,0,0],[a,a,0],[a,a,a]]),3);

$$\begin{bmatrix} a^8 & 0 & 0 \\ 8a^8 & a^8 & 0 \\ 36a^8 & 8a^8 & a^8 \end{bmatrix}$$

### 3.3 Beispiel: Resolventenkalkül

> restart: with(LinearAlgebra):

Gegeben sei ein festes Polynom:

> p:=x->x^3+a\*x^2+b\*x+c;

$$p := x \rightarrow x^3 + ax^2 + bx + c$$

Die zugehörige **Begleitmatrix** hat genau die Nullstellen des Polynoms als Eigenwerte:

> C:=CompanionMatrix(p(x),x);

$$C := \begin{bmatrix} 0 & 0 & -c \\ 1 & 0 & -b \\ 0 & 1 & -a \end{bmatrix}$$

> E:=IdentityMatrix(3);

$$E := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

Folgende Matrixfunktion bezeichnet man als **Resolvente** (singulär, d.h. undefiniert genau an den Eigenwerten von C):

> R:=z->(z\*E-C)^(-1);

$$R := z \rightarrow \frac{1}{zE - C}$$

>  $R(z);$

$$\begin{bmatrix} \frac{z^2 + za + b}{c + z^3 + z^2 a + zb} & -\frac{c}{c + z^3 + z^2 a + zb} & -\frac{cz}{c + z^3 + z^2 a + zb} \\ \frac{z + a}{c + z^3 + z^2 a + zb} & \frac{z(z + a)}{c + z^3 + z^2 a + zb} & -\frac{zb + c}{c + z^3 + z^2 a + zb} \\ \frac{1}{c + z^3 + z^2 a + zb} & \frac{z}{c + z^3 + z^2 a + zb} & \frac{z^2}{c + z^3 + z^2 a + zb} \end{bmatrix}$$

Mit derartigen Experimenten kann man 'vermuten', wie die Struktur im allgemeinen Fall aussieht -- Computeralgebra als Hilfsmittel für technische Beweisführungen...

Jetzt eine Prozedur:

Berechnung der Ableitung  $R'(z)$  der Matrixfunktion  $R(z) = (zI - A)^{-1}$   
(A eine beliebige gegebene quadratische Matrix)

... die Ableitung ist zu verstehen als Differentialquotient, elementweise (im Komplexen!)

$$R'(z) = \lim_{h \rightarrow 0} (R(z+h) - R(z))/h \quad \text{für } h > 0$$

```
> DRes:=proc(A::Matrix,z)
  local E,R,DR,h,i,j,n,zeta;
  uses LinearAlgebra; # lokal aktivieren
  if (RowDimension(A)<>ColumnDimension(A)) then
  >   error "A muss quadratische Matrix sein"
  > end if;
  n:=ColumnDimension(A);
  E:=IdentityMatrix(n);
  R:=zeta->(zeta*E-A)^(-1);
  DR:=(R(z+h)-R(z))/h;
  # den Limes kann man nur koeffizientenweise bilden:
  DR:=Matrix(
    [seq(
      [seq(simplify(limit(DR[i,j],h=0)),
        j=1..n)],i=1..n)];
  > end proc;
  > A:=Matrix([[1,2],[lambda,4]]);
```

$$A := \begin{bmatrix} 1 & 2 \\ \lambda & 4 \end{bmatrix}$$

>  $DRes(A, z);$

$$\begin{bmatrix} -\frac{z^2 - 8z + 16 + 2\lambda}{(z^2 - 5z + 4 - 2\lambda)^2} & -\frac{2(2z - 5)}{(z^2 - 5z + 4 - 2\lambda)^2} \\ -\frac{\lambda(2z - 5)}{(z^2 - 5z + 4 - 2\lambda)^2} & -\frac{z^2 - 2z + 1 + 2\lambda}{(z^2 - 5z + 4 - 2\lambda)^2} \end{bmatrix}$$

Ein kleiner 'Computerbeweis':

Wir verifizieren jetzt die Identität  $R'(z) = -R^2(z)$  für beliebige 2x2-Matrizen  
(vgl. skalaren Fall  $A=a$ )

> **A:=Matrix(2,2,symbol=a);**

$$A := \begin{bmatrix} a_{1,1} & a_{1,2} \\ a_{2,1} & a_{2,2} \end{bmatrix}$$

> **q1:=DRes(A,z);**

q1 :=

$$\left[ \left[ \begin{array}{l} -\frac{-2za_{2,2} + a_{1,2}a_{2,1} + z^2 + a_{2,2}^2}{(z^2 - za_{2,2} - a_{1,1}z + a_{1,1}a_{2,2} - a_{1,2}a_{2,1})^2}, \\ -\frac{a_{1,2}(2z - a_{2,2} - a_{1,1})}{(z^2 - za_{2,2} - a_{1,1}z + a_{1,1}a_{2,2} - a_{1,2}a_{2,1})^2}, \\ \left[ \begin{array}{l} -\frac{a_{2,1}(2z - a_{2,2} - a_{1,1})}{(z^2 - za_{2,2} - a_{1,1}z + a_{1,1}a_{2,2} - a_{1,2}a_{2,1})^2}, \\ -\frac{-2a_{1,1}z + a_{1,2}a_{2,1} + z^2 + a_{1,1}^2}{(z^2 - za_{2,2} - a_{1,1}z + a_{1,1}a_{2,2} - a_{1,2}a_{2,1})^2} \end{array} \right] \end{array} \right]$$

> **q2:=- (z\*IdentityMatrix(2)-A)^(-2);**

$$q2 := \left[ \left[ \begin{array}{l} -(-2za_{2,2} + a_{1,2}a_{2,1} + z^2 + a_{2,2}^2) / (4a_{1,1}z^2a_{2,2} + 2a_{1,1}za_{1,2}a_{2,1} - 2a_{1,1}z^3 - 2a_{1,1}za_{2,2} + a_{2,2}^2 + 2a_{1,2}a_{2,1}za_{2,2} + a_{1,2}^2a_{2,1}^2 - 2a_{1,2}a_{2,1}z^2 - 2z^3a_{2,2} + z^4 + z^2a_{2,2}^2 - 2a_{1,1}za_{2,2} + a_{1,1}^2z^2 + a_{1,1}^2a_{2,2}^2 - 2a_{1,2}a_{1,1}a_{2,1}a_{2,2}), \\ -(a_{1,2}(2z - a_{2,2} - a_{1,1})) / (4a_{1,1}z^2a_{2,2} + 2a_{1,1}za_{1,2}a_{2,1} - 2a_{1,1}z^3 - 2a_{1,1}za_{2,2} + a_{2,2}^2 + 2a_{1,2}a_{2,1}za_{2,2} + a_{1,2}^2a_{2,1}^2 - 2a_{1,2}a_{2,1}z^2 - 2z^3a_{2,2} + z^4 + z^2a_{2,2}^2 - 2a_{1,1}za_{2,2} + a_{1,1}^2z^2 + a_{1,1}^2a_{2,2}^2 - 2a_{1,2}a_{1,1}a_{2,1}a_{2,2}), \\ -(a_{2,1}(2z - a_{2,2} - a_{1,1})) / (4a_{1,1}z^2a_{2,2} + 2a_{1,1}za_{1,2}a_{2,1} - 2a_{1,1}z^3 - 2a_{1,1}za_{2,2} + a_{2,2}^2 + 2a_{1,2}a_{2,1}za_{2,2} + a_{1,2}^2a_{2,1}^2 - 2a_{1,2}a_{2,1}z^2 - 2z^3a_{2,2} + z^4 + z^2a_{2,2}^2 - 2a_{1,1}za_{2,2} + a_{1,1}^2z^2 + a_{1,1}^2a_{2,2}^2 - 2a_{1,2}a_{1,1}a_{2,1}a_{2,2}), \\ -(-2a_{1,1}z + a_{1,2}a_{2,1} + z^2 + a_{1,1}^2) / (4a_{1,1}z^2a_{2,2} + 2a_{1,1}za_{1,2}a_{2,1} - 2a_{1,1}z^3 - 2a_{1,1}za_{2,2} + a_{2,2}^2 + 2a_{1,2}a_{2,1}za_{2,2} + a_{1,2}^2a_{2,1}^2 - 2a_{1,2}a_{2,1}z^2 - 2z^3a_{2,2} + z^4 + z^2a_{2,2}^2 - 2a_{1,1}za_{2,2} + a_{1,1}^2z^2 + a_{1,1}^2a_{2,2}^2 - 2a_{1,2}a_{1,1}a_{2,1}a_{2,2}) \end{array} \right] \right]$$

> **simplify(q1-q2); # hier simplify erforderlich!**

$$\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$$

Dasselbe für 3x3-Matrizen:

> **A:=Matrix(3,3,symbol=a);**

$$A := \begin{bmatrix} a_{1,1} & a_{1,2} & a_{1,3} \\ a_{2,1} & a_{2,2} & a_{2,3} \\ a_{3,1} & a_{3,2} & a_{3,3} \end{bmatrix}$$

```
> q1:=DRes(A,z):
```

```
> q2:=-(z*IdentityMatrix(3)-A)^(-2):
```

```
> simplify(q1-q2);
```

$$\begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

... wird für höhere Dimensionen mühsam (Laufzeit!)

```
> restart: with(LinearAlgebra):
```

### 3.4 Abschließendes Beispiel: Cauchy-Integral

Für komplex differenzierbare Funktionen  $f(z)$  gilt die **Cauchy'sche Integralformel** der Funktionentheorie:

$f(z)$  = (komplexes Kurvenintegral von  $f(\zeta)/(\zeta-z) d\zeta$ ) /  $(2 i \pi)$   
entlang einer geschlossenen Kurve, die  $z$  im Inneren enthält

Beispiel:  $f(z)=e^z$ , Kurve = Kreis mit Radius  $r>0$  mit  $z$  als Mittelpunkt

Für die Berechnung setzen wir  $\zeta = \zeta(\phi) = z + r e^{i \phi}$  (Substitution),  
mit  $d\zeta = i r e^{i \phi} d\phi$ ,

also:

```
> f:=exp; z:='z':
```

```
  r:='r':assume(r>0); phi:='phi':assume(phi,real);
```

```
  f:=exp
```

```
> zeta:=phi->z+r*exp(I*phi);
```

$$\zeta := \phi \rightarrow z + r e^{i\phi}$$

```
> CauchyIntegral:=(1/(2*I*Pi))*Int((f(zeta(phi))/(zeta(phi)-z))
```

```
  *I*r*exp(I*phi),phi=0..2*Pi);
```

$$\text{CauchyIntegral} := -\frac{\frac{1}{2} I \int_0^{2\pi} I e^z + r e^{I\phi} d\phi}{\pi}$$

*With assumptions on r, phi*

```
> value(CauchyIntegral); # forciert Auswertung Int->int !
```

```
  ez
```

... stimmt tatsächlich ...

Jetzt dasselbe für **Matrix-wertige** Funktion, z.B. die **Potenz  $A^k$** :

Achtung: Alle **Eigenwerte von A** müssen innerhalb der Kurve liegen!

```
> k:=5;
```

```
  k:=5
```

```
> f:=z->z^k: f(z);
```

```
  z5
```

```
> n:=2; A:=Matrix([[1,2],[3,4]]); E:=IdentityMatrix(n):
```

```
n := 2
```

$$A := \begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix}$$

```
> f(A);
```

$$\begin{bmatrix} 1069 & 1558 \\ 2337 & 3406 \end{bmatrix}$$

```
> evalf(Eigenvalues(A));
```

$$\begin{bmatrix} 5.37228132326901432993 \\ -0.37228132326901432993 \end{bmatrix}$$

... also wähle z.B. Kreis mit Mittelpunkt 0 und Radius r=6

```
> r:=6: zeta:=phi->r*exp(I*phi);
```

```
CauchyIntegral:=(i,j)->
```

```
(1/(2*I*Pi))
```

```
*Int((f(zeta(phi))*I*r*exp(I*phi)
```

```
*((zeta(phi)*E-A)^(-1))[i,j]),phi=0..2*Pi):
```

$$\zeta := \phi \rightarrow r e^{I\phi}$$

```
> value(CauchyIntegral(1,2)); simplify(%); # O.K.
```

$$-\frac{1558 I (383 I - 235 \sqrt{119})}{383 + 235 I \sqrt{119}}$$
$$1558$$

Die ganze Ergebnismatrix:

```
> Matrix(2,2,(i,j)->simplify(value(CauchyIntegral(i,j)))), f(A);
```

$$\begin{bmatrix} 1069 & 1558 \\ 2337 & 3406 \end{bmatrix}, \begin{bmatrix} 1069 & 1558 \\ 2337 & 3406 \end{bmatrix}$$

Also OK. Müsste man noch effizienter programmieren; die Integrale werden für höhere Dimensionen ziemlich 'wild'.

... Hier stößt man an die Grenze dessen, was Maple noch beherrscht.

Anmerkung: Das Cauchy-Beispiel soll nur die Fähigkeiten bzw. Grenzen von Maple in der Matrizenanalyse aufzeigen. Das Cauchy-Integral selbst dient i.Allg. nicht zum Rechnen, sondern es ist ein 'Workhorse' in der Matrixtheorie, z.B. zur Abschätzung der Werte nicht elementar berechenbarer Matrixfunktionen.

## 4. Spezielle Datenstrukturen

Der **Matrix**-Typ ist von `rtables` abgeleitet und unterstützt verschiedenste spezielle Matrixstrukturen und damit zusammenhängend spezielle Speichermuster über sogenannte 'indexing functions' - relevant für große, speziell strukturierte Matrizen (ähnlich für Vektoren). Einige Beispiele:

```
> ? shape
```

Meist wird der 'storage mode' aus dem 'shape' abgeleitet.

```
> A:=Matrix(3,3,symbol=a,shape=symmetric);
```

$$A := \begin{bmatrix} a_{1,1} & a_{1,2} & a_{1,3} \\ a_{1,2} & a_{2,2} & a_{2,3} \\ a_{1,3} & a_{2,3} & a_{3,3} \end{bmatrix}$$

```
> op(A);
```

```
3, 3, {(1, 1) = a1,1, (1, 2) = a1,2, (1, 3) = a1,3, (2, 2) = a2,2, (2, 3) = a2,3, (3, 3) = a3,3}, datatype  
= anything, storage = triangular_upper, order = Fortran_order, shape = [symmetric]
```

```
> A[2,1]:=2; A;
```

$$A_{2,1} := 2$$
$$A := \begin{bmatrix} a_{1,1} & 2 & a_{1,3} \\ 2 & a_{2,2} & a_{2,3} \\ a_{1,3} & a_{2,3} & a_{3,3} \end{bmatrix}$$

```
> A:=Matrix(4,4,symbol=a,shape=band[1]); # tridiagonal
```

$$A := \begin{bmatrix} a_{1,1} & a_{1,2} & 0 & 0 \\ a_{2,1} & a_{2,2} & a_{2,3} & 0 \\ 0 & a_{3,2} & a_{3,3} & a_{3,4} \\ 0 & 0 & a_{4,3} & a_{4,4} \end{bmatrix}$$

```
> A[1,4]:=1;
```

```
Error, attempt to assign a value outside Matrix bands
```

```
> op(A);
```

```
4, 4, {(1, 1) = a1,1, (1, 2) = a1,2, (2, 1) = a2,1, (2, 2) = a2,2, (2, 3) = a2,3, (3, 2) = a3,2, (3, 3) = a3,3,  
(3, 4) = a3,4, (4, 3) = a4,3, (4, 4) = a4,4}, datatype = anything, storage = band1,1, order  
= Fortran_order, shape = [band1,1]
```

```
> A:=Matrix(5,5,storage=sparse);
```

$$A := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

```
> A[1,1]:=1; op(A);
```

$$A_{1,1} := 1$$

```
5, 5, {(1, 1) = 1}, datatype = anything, storage = sparse, order = Fortran_order, shape = [ ]
```

## ▼ 5. Ergänzung: packages numapprox und plots

Neben der Numerischen Linearen Algebra (**LinearAlgebra**) enthält


das **numapprox** - package einige für die Numerische Mathematik relevante Tools.

```
> with(numapprox);  
[chebdeg, chebmult, chebpade, chebsort, chebyshev, confracform, hermite_pade, hornerform, infnorm,  
  laurent, minimax, pade, remez]  
  
> # z.B.: maximum-norm einer Funktion:  
> infnorm(sin,0..1);  
  
0.84147098480789650665
```

Das package **plots** enthält diverse Funktionen für grafische Darstellung.


Einige Beispiele:

```
> with(plots);  
[animate, animate3d, animatecurve, arrow, changecoords, complexplot, complexplot3d, conformal,  
  conformal3d, contourplot, contourplot3d, coordplot, coordplot3d, densityplot, display, dualaxisplot,  
  fieldplot, fieldplot3d, gradplot, gradplot3d, graphplot3d, implicitplot, implicitplot3d, inequal,  
  interactive, interactiveparams, intersectplot, listcontplot, listcontplot3d, listdensityplot, listplot,  
  listplot3d, loglogplot, logplot, matrixplot, multiple, odeplot, pareto, plotcompare, pointplot,  
  pointplot3d, polarplot, polygonplot, polygonplot3d, polyhedra_supported, polyhedraplot, rootlocus,  
  semilogplot, setcolors, setoptions, setoptions3d, spacecurve, sparsematrixplot, surfdata, textplot,  
  textplot3d, tubeplot]  
  
> plot1:=plot3d(x+y,x=-1..1,y=-1..1,color=red); # Funktion, deren  
Ausführung den Plot bewirkt  
  
# (interne  
Datenstruktur für Grafikdaten)  
plot1 := PLOT3D(...)  
  
> whattype(plot1);  
function  
  
> plot2:=implicitplot3d(x^2+y^2+z^2=1,x=-1..1,y=-1..1,z=-1..1);  
plot2 := PLOT3D(...)  
  
> display(plot1,plot2,axes=boxed,scaling=constrained);
```




```
> animate((sin(t*x),x=-1..1),t=0..10,thickness=8);
```






```
> animate3d((sin(t*x*y),x=-1..1,y=-1..1),t=0..10,shading=zhue,axes=boxed);
```



```
> contourplot(sin(x*y),x=-1..1,y=-1..1,thickness=4,axes=boxed,  
contours=20);
```



> ### und vieles Weitere!

> ### Beachte Kontextmenü und Exportmöglichkeiten.

> #####  
#####